VBA Code Sample #33 ( Selecting Outlook Contacts for a Word Merge Letter from a Multi-Select Listbox

Created 7-10-2000

Helen Feddema (hfeddema@hvc.rr.com)

Introduction

This code sample shows how to create a UserForm that is popped up when a new document is created from a Word template.  The UserForm has a multi-select listbox to display contacts from an Outlook Contacts folder, selected from a Select Folder dialog.  The contacts selected in the listbox are used to generate letters to a set of contacts.

Technically, this template is not a merge letter, such as the ones you create from the Mail Merge selection on the Word Tools menu.  Nor does it use Word document properties to store the data from Outlook, as I do in several other code samples (that method requires a template with the appropriate document properties prepared in advance).  Instead, this method uses the TypeText method to write the data from the Outlook contacts directly into the newly created Word letters, using the standard Normal.dot template to create the letters.

I prefer to avoid Word merge letters, as they are extremely memory-intensive, and in the case of Outlook, your only choice is to create a merge letter to all the contacts in a folder, which in the case of the Contacts folder might be a very large merge indeed.  If you need to select specific contacts from a folder to receive a set of letters, you need a way of choosing the contacts, and that is what the multi-select listbox accomplishes.

The following steps guide you through creating a Word UserForm with a multi-select listbox; if you prefer to just view the finished product, open the Outlook Contact Letter.dot (right-click in the Explorer and select Open; don’t double-click, as that will create a new letter based on the template).  Switch to the VBA window by pressing Alt-F11, open the Forms folder under the Outlook Contact Letter TemplateProject in the Project Explorer, and double-click the frmOutlookData form to inspect the UserForm.

Instructions

To create a UserForm with a Folder selector command button and a multi-select listbox from scratch, follow the steps below:

1. Create a new Word document.

2. Press Alt-F11 or select Tools|Macro|Visual Basic Editor to open the VBA window.

3. The new document will appear as Document2 (or some other number) in the Project Explorer.  Since it is a new document, it has no Forms folder.

4. Select the Documentn project, and click the Insert UserForm button in the VBA window toolbar to create a new UserForm for the project.

5. The UserForm appears, under a new Forms folder.  It is initially named UserForm1.

6. First rename the UserForm by opening the Properties pane (if it is not already open), and changing UserForm1 to frmOutlookData in the (Name) row, which is the top row in the Properties pane.

7. Set the UserForm’s Height property to 415, and its Width property to 356 (these are approximate values; you may wish to make the form smaller, especially if you have a small monitor).

8. Click the UserForm to make the Toolbox appear; if it doesn’t appear, drop down the View menu and select Toolbox.

9. Place two Label controls, two TextBox controls, a Listbox control, and three CommandButton controls on the UserForm, by clicking the corresponding tools in the Toolbox.  On my UserForm, these controls have the following name, size and position properties; you may adjust them as needed to fit on a UserForm of your preferred size:

Label Controls

Name:
lblSelectContacts

Caption:
Select Contacts:

Height:
12

Left:
18

Top:
66

Width:
66

Name:
lblLetterText

Caption:
Letter Text:

Height:
12

Left:
18

Top:
249

Width:
66

TextBox Controls

Name:
txtFolder

Height:
18

Left:
96

Top:
12

Width:
240

Name:
txtLetterText

Height:
102

Left:
96

Top:
246

Width:
240

CommandButton Controls

Name:
cmdSelectFolder

Caption:
Select Folder

Height:
18

Left:
18

Top:
12

Width:
72

Name:
cmdCancel

Cancel:
True

Caption:
Cancel

Height:
18

Left:
180

Top:
366

Width:
54

Name:
cmdMerge

Caption:
Merge

Height:
18

Left:
276

Top:
366

Width:
54

10. The Listbox control has a number of special properties that must be set to make it a multi-select listbox.  If you are familiar with Access listboxes, note that the properties are slightly different for listboxes on Office UserForms.  There are also differences in coding, which will be discussed in the inline code comments in the Code section.

ListBox Control

Name:
lstContacts

ColumnCount:
4

ColumnWidths:
75 pt;75 pt;0 pt;0 pt

Height:
168

Left:
96

ListStyle:
1 – fmListStyleOption

MultiSelect:
2 - fmMultiSelectExtended

Top:
66

Width:
240

11. The finished UserForm is shown below, in design view.

12. Now that the UserForm has all the necessary controls, the next step is to write the code.  You need code behind the UserForm, in several procedures, plus a Document_New procedure in the oddly-named ThisDocument object, which contains code for several document events.

13. The ThisDocument object is located under the Microsoft Word Objects folder for Documentn.  Double-click this object to open it, then select Document from the Object box (on the left side).  A code stub for the Document_New procedure appears; enter the lines shown below to open the frmOutlookContacts UserForm when a new document is created from the template (I have also included an error handler in the procedure):

Private Sub Document_New()

On Error GoTo ErrorHandler

   Load frmOutlookData

   frmOutlookData.Show

ErrorHandlerExit:

   Exit Sub

ErrorHandler:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume ErrorHandlerExit

End Sub

14. The UserForm requires several procedures:  a cmdSelectFolder_Click procedure to select an Outlook contacts folder and fill the listbox with contact data from the folder, a cmdMerge_Click procedure to create the Word letters with the Outlook data and the letter text entered in the textbox on the UserForm, and a cmdCancel_Click procedure to close the UserForm.  The code is broken up into segments, with descriptive text in between (see the Code section below).

15. To open the UserForm’s code window, select frmOutlookData in the Project Explorer, right-click it and select View Code from the context menu.

Using the template

1. Make a new document from the Outlook Contact Letter.dot template by double-clicking it in the Explorer, or selecting it as the template to use when creating a new document from within Word.

2. When the UserForm appears, click the Select Folder command button.

3. A Select Folder dialog opens.

[image: image7.png]Select Folder

Folders
B Windlows 98 OSR A] oK
B Windaws CE
© Calendar Cancel
€ Class Dates
G Computer Hardware New.
G Conferences J
@@ Contacts

© Cantacts to Export
©cr
@ Custom Contacts
& (7 Deleted ltems (43)
2 Design Schemes Beta
3 Developers Resources
€3 Development

3 Nevalnnment dnimal





4. Select a folder containing contacts and click the OK button; the Select Contacts listbox will be filled with contact data from the folder.

5. As the contact items in the folder are processed, you may get a message like the one below, for contacts using non-standard forms.  Just click the Disable Macros button to proceed.

6. Click one or more contact names (use Shift-Click to highlight a group of contiguous names, or Ctrl-Click to highlight non-contiguous names), then click the Merge button to create the letters.  A typical letter is shown below.

Code

Option Explicit

Declare the lst variable in the Declarations section, since it is used in several procedures

Dim lst As MSForms.ListBox

Private Sub cmdSelectFolder_Click()

On Error GoTo ErrorHandler

Declare Outlook variables

   Dim appOutlook As New Outlook.Application

   Dim nms As Outlook.NameSpace

   Dim fld As Outlook.MAPIFolder

   Dim itms As Outlook.Items

Declare variables needed to work with listbox and the array used to fill it

   Dim strContactArray() As String

   Dim lngContactCount As Long

   Dim itm As Object

   Dim lngRow As Long

Open Select Folder dialog using the PickFolder method of the NameSpace object, so user can select a Contacts folder

SelectFolder:

   Set nms = appOutlook.GetNamespace("MAPI")

   Set fld = nms.PickFolder

   If fld Is Nothing Then

      MsgBox "Please select a folder"

      Exit Sub

   Else

Test whether folder is a Contacts folder

      If fld.DefaultItemType <> olContactItem Then

         MsgBox "Please select a Contacts folder"

         GoTo SelectFolder

      End If

   End If

Put folder name into txtFolder, and fill txtLetterText with default text to be replaced by the user

   Me.txtFolder.Value = fld.Name

   Me![txtLetterText].Value = "Type letter text here"

   Set itms = fld.Items

   lngContactCount = itms.Count

   itms.Sort "[LastName]", False

   Set lst = lstContacts

   lngContactCount = itms.Count

   If lngContactCount = 0 Then

      MsgBox "No contacts found in selected folder"

      Exit Sub

   Else

      Debug.Print lngContactCount & " unfiltered contacts"

   End If

Count number of contacts with basic name and address data

   lngRow = 0

Check each item in the folder and only process contact items

   For Each itm In itms

      If itm.Class = olContact Then

         If itm.FullName <> "" And itm.CompanyName <> "" _

            And itm.BusinessAddress <> "" Then

            lngRow = lngRow + 1

         End If

      End If

   Next

Fill Contacts array with data from each contact in folder that has basic name and address data

   ReDim strContactArray(lngRow - 1, 3)

   lngRow = 0

   For Each itm In itms

      If itm.Class = olContact Then

         If itm.FullName <> "" And itm.CompanyName <> "" _

            And itm.BusinessAddress <> "" Then

            'Debug.Print "Adding " & itm.LastNameAndFirstName & " to array"

            strContactArray(lngRow, 0) = itm.LastNameAndFirstName

            strContactArray(lngRow, 1) = itm.CompanyName

            strContactArray(lngRow, 2) = itm.BusinessAddress

            strContactArray(lngRow, 3) = itm.FullName

            lngRow = lngRow + 1

         End If

      End If

   Next

   Debug.Print lngRow & " filtered contacts"

Fill the listbox with data from the array

   lst.List() = strContactArray

ErrorHandlerExit:

   Exit Sub

ErrorHandler:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume ErrorHandlerExit

End Sub

Private Sub cmdMerge_Click()

On Error GoTo ErrorHandler

Declare listbox variables

   Dim intItem As Integer

   Dim intIndex As Integer

   Dim intRow As Integer

   Dim intRows As Integer

   Dim intColumn As Integer

   Dim intColumns As Integer

   Dim strLetterText As String

   Set lst = Me![lstContacts]

   strLetterText = Me![txtLetterText]

   intColumns = lst.ColumnCount

   intRows = lst.ListCount - 1

Unlike Access, which has the handy ItemsSelected collection of a listbox to work with, when working with the selected items in a listbox on an Office UserForm, you have to check the Selected property for each item in the list, in order to determine if it has been selected

   For intItem = 1 To intRows

      If lst.Selected(intItem) = True Then

Open a new letter based on the Normal template

         'Debug.Print "Creating a letter to : " & Nz(lst.Column(3, intItem))

         Application.Documents.Add

Paste Outlook data to letter for each selected contact

         With Selection

            .ParagraphFormat.Alignment = wdAlignParagraphRight

Insert current date in a specific format, as text

            .InsertDateTime DateTimeFormat:="dddd, MMMM dd, yyyy", _

               InsertAsField:=False

            .TypeParagraph

            .ParagraphFormat.Alignment = wdAlignParagraphLeft

            .TypeParagraph

            .TypeParagraph

            .TypeText Text:=Nz(lst.Column(3, intItem))

            .TypeParagraph

            .TypeText Text:=Nz(lst.Column(1, intItem))

            .TypeParagraph

            .TypeText Text:=Nz(lst.Column(2, intItem))

            .TypeParagraph

            .TypeParagraph

            .TypeText "Dear Sir:"

            .TypeParagraph

            .TypeParagraph

            .TypeText strLetterText

            .TypeParagraph

            .TypeParagraph

            .TypeText "Yours sincerely"

            .TypeParagraph

            .TypeParagraph

            .TypeParagraph

            .TypeParagraph

Insert Author value from Word as the letter signature

            .TypeText Application.ActiveDocument.BuiltInDocumentProperties(wdPropertyAuthor)

         End With

      End If

   Next intItem

Close the UserForm

   Unload Me

ErrorHandlerExit:

   Exit Sub

ErrorHandler:

   MsgBox "Error No: " & Err.Number & "; Description: " & Err.Description

   Resume ErrorHandlerExit

End Sub

Private Sub cmdCancel_Click()

Close the UserForm

   Unload frmOutlookData

End Sub

Notes

· To declare variables as specific class types (instead of just as Object), you need to have a reference set to the appropriate object library.  The references in the template should be correct, but in case you have a problem (indicated by a message such as "ActiveX component can't create object" or "Can't open macro storage"), open the Word VBA window, drop down the Tools menu and select References; check that the appropriate version of the Outlook object library is checked (Outlook 98 for Word 97, or Outlook 9 for Word 2000).  You also need a reference to the Access object library, to support the use of the Nz function.  The first screen shot below shows the references for Office 97 (with Outlook 98); the second shows the references for Office 2000.

· In order to use this technique on a template of your own, first import the UserForm files by placing the frmOutlookData files on your computer.  Open your template, switch to the VBA window, and drop the File menu and select Import file, and select frmOutlookData.frm (the code in the frmOutlookData.frx file will be imported along with the form definition).  You will also need to copy the Document_New procedure into the ThisDocument object for the template.

Contents of Zip File

	File Name
	File Type
	Destination Folder

	Outlook Contacts Letter.dot
	Word 97/2000 template
	\Templates (or wherever you want)

	frmOutlookData.frm, frmOutlookData.frx
	UserForm files
	\My Documents  (or wherever you want)

	VB Script Code Sample #33 -- Selecting Outlook Contacts for a Word Merge Letter from a Multi-Select Listbox
	Word 97 document
	\My Documents  (or wherever you want)

